Undetermined Coefficients Examples

Example: Let's solve the following non-homogeneous linear differential equation (with constant coefficients) using the method of undetermined coefficients:

```
> restart;
> diff('y(t)',t$3)-3*diff('y(t)',t$2)+4*'y(t)'=18*exp(2*t)+54*t*exp(2*t);
```

$$\frac{d^3}{dt^3} y(t) - 3\left(\frac{d^2}{dt^2} y(t)\right) + 4\, y(t) = 18\, e^{2t} + 54\, t\, e^{2t} \quad \textbf{(1)}$$

First, we'll write down the characteristic polynomial and solve the corresponding homogeneous equation.

```
> solve(r^3-3*r^2+4=0);
```

$$-1, 2, 2 \quad \textbf{(2)}$$

```
> y := C_1*exp(-t)+C_2*exp(2*t)+C_3*t*exp(2*t);
```

$$y := C\_1\, e^{-t} + C\_2\, e^{2t} + C\_3\, t\, e^{2t} \quad \textbf{(3)}$$

So we have $L = (D+1)(D-2)^2$ and $(D-2)^2$ annihilates $18e^{2t} + 54\, t\, e^{2t}$. Normally our template for a particular solution would be $Ae^{2t} + B\, t\, e^{2t}$, but this overlaps with the homogeneous solution, so it must be adjusted (multiplying by t^2 makes sure it no longer overlaps): $y_p = At^2e^{2t} + Bt^3e^{2t}$ is our template solution.

```
> y_p := A*t^2*exp(2*t)+B*t^3*exp(2*t);
```

$$y\_p := A\, t^2\, e^{2t} + B\, t^3\, e^{2t} \quad \textbf{(4)}$$

Let's plug in our template, simplify, set coefficients of linearly independent functions equal to each other, and solve the corresponding linear system...

```
> simplify(diff(y_p,t$3)-3*diff(y_p,t$2)+4*y_p=18*exp(2*t)+54*t*exp(2*t));
```

$$6\, e^{2t}\, (A + B + 3\, B\, t) = 18\, e^{2t}\, (1 + 3\, t) \quad \textbf{(5)}$$

```
> solve({6*A+6*B=18, 18*B=54});
```

$$\{A = 0, B = 3\} \quad \textbf{(6)}$$

Our general solution is $y = C_1\, e^{-t} + C_2\, e^{2t} + C_3\, t\, e^{2t} + 3\, t^3 e^{2t}$.

Example: Let's solve the following non-homogeneous linear differential equation (with constant coefficients) using the method of undetermined coefficients:

```
> restart;
> diff('y(t)',t$2)+2*diff('y(t)',t)+10*'y(t)' = -2*exp(-t)*sin(3*t)+3*cos(5*t);
```

$$\frac{d^2}{dt^2}\,y(t) + 2\left(\frac{d}{dt}\,y(t)\right) + 10\,y(t) = -2\,e^{-t}\sin(3\,t) + 3\cos(5\,t) \qquad \textbf{(7)}$$

First, we'll write down the characteristic polynomial and solve the corresponding homogeneous equation.

```
> solve(r^2+2*r+10=0);
```

$$-1+3\,I,\ -1-3\,I \qquad \textbf{(8)}$$

```
> y := C_1*exp(-t)*cos(3*t)+C_2*exp(-t)*sin(3*t);
```

$$y := C\_1\,e^{-t}\cos(3\,t) + C\_2\,e^{-t}\sin(3\,t) \qquad \textbf{(9)}$$

Normally our template for a particular solution would be $Ae^{-t}\cos(3\,t) + Be^{-t}\sin(3\,t) + C\sin(5\,t) + E\cos(5\,t)$, but there is some overlap with the homogeneous solution, so it must be adjusted (multiplying the first two terms *t* makes sure it no longer overlaps): $y_p = A\,t\,e^{-t}\cos(3\,t) + B\,t\,e^{-t}\sin(3\,t) + C\sin(5\,t) + E\cos(5\,t)$ is our template solution. [Note: I'm avoiding using "D" as an undetermined coefficient since Maple has reserved the symbol "D" for other uses.]

```
> y_p := A*t*exp(-t)*cos(3*t)+B*t*exp(-t)*sin(3*t)+C*sin(5*t)+E*cos(5*t);
```

$$y\_p := A\,t\,e^{-t}\cos(3\,t) + B\,t\,e^{-t}\sin(3\,t) + C\sin(5\,t) + E\cos(5\,t) \qquad \textbf{(10)}$$

Let's plug in our template, simplify, set coefficients of linearly independent functions equal to each other, and solve the corresponding linear system...

```
> simplify(diff(y_p,t$2)+2*diff(y_p,t)+10*y_p=-2*exp(-t)*sin(3*t)+3*cos(5*t));
```

$$-6\,A\,e^{-t}\sin(3\,t) + 6\,B\,e^{-t}\cos(3\,t) - 15\,C\sin(5\,t) - 15\,E\cos(5\,t) + 10\,C\cos(5\,t) - 10\,E\sin(5\,t) = -2\,e^{-t}\sin(3\,t) + 3\cos(5\,t) \qquad \textbf{(11)}$$

```
> solve({-6*A=-2, 6*B=0, -15*C-10*E=0, -15*E+10*C=3});
```

$$\left\{A = \frac{1}{3},\ B = 0,\ C = \frac{6}{65},\ E = -\frac{9}{65}\right\} \qquad \textbf{(12)}$$

Our general solution is

$$y = C_1\,e^{-t}\cos(3\,t) + C_2\,e^{-t}\sin(3\,t) + \frac{1}{3}t\,e^{-t}\cos(3\,t) + \frac{6}{65}\sin(5\,t) - \frac{9}{65}\cos(5\,t).$$

Example: Let's solve the following non-homogeneous Cauchy-Euler linear differential equation using the method of undetermined coefficients:

```
> restart;
```

```
> t^2*diff('y(t)',t$2)+5*t*diff('y(t)',t)+4*'y(t)' = -10*sin(3*ln(t))
  +24*cos(3*ln(t));
```

$$t^2 \left(\frac{d^2}{dt^2} y(t)\right) + 5\,t \left(\frac{d}{dt} y(t)\right) + 4\,y(t) = -10 \sin(3 \ln(t)) + 24 \cos(3 \ln(t)) \qquad \textbf{(13)}$$

First, we'll write down the characteristic polynomial and solve the corresponding homogeneous equation.

```
> solve(r*(r-1)+5*r+4=0);
```

$$-2, -2 \qquad \textbf{(14)}$$

```
> y := C_1*t^(-2)+C_2*ln(t)*t^(-2);
```

$$y := \frac{C\_1}{t^2} + \frac{C\_2 \ln(t)}{t^2} \qquad \textbf{(15)}$$

Our first try at a template for a particular solution should be $A \cos(3 \ln(t)) + B \sin(3 \ln(t))$. Since there's no overlap, we don't need to adjust our template.

```
> y_p := A*cos(3*ln(t))+B*sin(3*ln(t));
```

$$y\_p := A \cos(3 \ln(t)) + B \sin(3 \ln(t)) \qquad \textbf{(16)}$$

Let's plug in our template, simplify, set coefficients of linearly independent functions equal to each other, and solve the corresponding linear system...

```
> simplify(t^2*diff(y_p,t$2)+5*t*diff(y_p,t)+4*y_p = -10*sin(3*ln(t))
  +24*cos(3*ln(t)));
```

$$-5\,A \cos(3 \ln(t)) - 12\,A \sin(3 \ln(t)) - 5\,B \sin(3 \ln(t)) + 12\,B \cos(3 \ln(t)) = -10 \sin(3 \ln(t)) + 24 \cos(3 \ln(t)) \qquad \textbf{(17)}$$

```
> solve({-5*A+12*B=24, -12*A-5*B=-10});
```

$$\{A = 0, B = 2\} \qquad \textbf{(18)}$$

Our general solution is $y = C_1 t^{-2} + C_2 \ln(t) t^{-2} + 2 \sin(3 \ln(t))$

Example: Let's solve the following non-homogeneous Cauchy-Euler linear differential equation using the method of undetermined coefficients:

```
> restart;
> t^2*diff('y(t)',t$2)+5*t*diff('y(t)',t)+4*'y(t)'=8*t^(-2);
```

$$t^2 \left(\frac{d^2}{dt^2} y(t)\right) + 5\,t \left(\frac{d}{dt} y(t)\right) + 4\,y(t) = \frac{8}{t^2} \qquad \textbf{(19)}$$

Since the corresponding homogeneous equation is the same as the last example, we have...

```
> y := C_1*t^(-2)+C_2*ln(t)*t^(-2);
```

(20)

$$y := \frac{C\_1}{t^2} + \frac{C\_2 \ln(t)}{t^2} \quad \textbf{(20)}$$

Our first try at a template for a particular solution is $y_p = A\,t^{-2}$. However, this overlaps with the homogeneous solution so we must adjust by multiplying by some power of $\ln(x)$. In this case, we need $(\ln(x))^2$. This gives us the template solution $y_p = A\,t^{-2}(\ln(t))^2$.

```
> y_p := A*t^(-2)*(ln(t))^2;
```

$$y\_p := \frac{A \ln(t)^2}{t^2} \quad \textbf{(21)}$$

Let's plug in our template, simplify, set coefficients of linearly independent functions equal to each other, and solve the corresponding linear system...

```
> simplify(t^2*diff(y_p,t$2)+5*t*diff(y_p,t)+4*y_p = 8*t^(-2));
```

$$\frac{2\,A}{t^2} = \frac{8}{t^2} \quad \textbf{(22)}$$

```
> solve({2*A=8});
```

$$\{A = 4\} \quad \textbf{(23)}$$

Our general solution is $y = C_1 t^{-2} + C_2 \ln(t) t^{-2} + 4\,(\ln(t))^2\,t^{-2}$.

<u>Example:</u> Find $\int t\,e^{-t} \cos(3\,t)\;dt$ using the method of undetermined coefficients.

Finding this integral is the same as solving $y' = t\,e^{-t}\cos(3\,t)$. Our template for a solution should be $y_p = (At + B)e^{-t}\cos(3\,t) + (Ct + E)e^{-t}\sin(3\,t)$. We need to differentiate, equate coefficients, and solve some linear equations...

```
> restart;
> y_p := (A*t+B)*exp(-t)*cos(3*t)+(C*t+E)*exp(-t)*sin(3*t);
```

$$y\_p := (A\,t + B)\,e^{-t}\cos(3\,t) + (C\,t + E)\,e^{-t}\sin(3\,t) \quad \textbf{(24)}$$

```
> simplify(diff(y_p,t)=t*exp(-t)*cos(3*t));
```

$$-e^{-t}\,(-A\cos(3\,t) + \cos(3\,t)\,A\,t + \cos(3\,t)\,B + 3\sin(3\,t)\,A\,t + 3\sin(3\,t)\,B - C\sin(3\,t) + \sin(3\,t)\,C\,t + \sin(3\,t)\,E - 3\cos(3\,t)\,C\,t - 3\cos(3\,t)\,E) = t\,e^{-t}\cos(3\,t) \quad \textbf{(25)}$$

```
> solve({-A+3*C=1,A-B+3*E=0,-3*A-C=0,-3*B+C-E=0});
```

$$\left\{A = -\frac{1}{10}, B = \frac{2}{25}, C = \frac{3}{10}, E = \frac{3}{50}\right\} \quad \textbf{(26)}$$

Thus $\int t\,e^{-t}\cos(3\,t)\;dt = \left(-\frac{1}{10}t + \frac{2}{25}\right)e^{-t}\cos(3\,t) + \left(\frac{3}{10}t + \frac{3}{50}\right)e^{-t}\sin(3\,t) + C$ (Note:

$y = C$ is our homogeneous solution.)

Of course, if we're going to use Maple anyway, it would be better just to go ahead and integrate...

```
> int(t*exp(-t)*cos(3*t),t);
```

$$\left(-\frac{1}{10}t+\frac{2}{25}\right)e^{-t}\cos(3t)-\left(-\frac{3}{10}t-\frac{3}{50}\right)e^{-t}\sin(3t) \quad \textbf{(27)}$$

Example: Find $\int (\ln(t))^2 t^{-\frac{7}{2}}\, dt$ using the method of undetermined coefficients.

Find this integral is equivalent to solving $t\,y' = (\ln(t))^2\, t^{-\frac{5}{2}}$ (notice we had to multiply both sides by t).

Our template for a solution should be $y_p = A(\ln(t))^2\, t^{-\frac{5}{2}} + B(\ln(t))\, t^{-\frac{5}{2}} + C\, t^{-\frac{5}{2}}$. We need to differentiate, equate coefficients, and solve some linear equations...

```
> restart;
> y_p := A*(ln(t))^2*t^(-5/2)+B*ln(t)*t^(-5/2)+C*t^(-5/2);
```

$$y\_p := \frac{A\ln(t)^2}{t^{5/2}}+\frac{B\ln(t)}{t^{5/2}}+\frac{C}{t^{5/2}} \quad \textbf{(28)}$$

```
> simplify(t*diff(y_p,t)=(ln(t))^2*t^(-5/2));
```

$$-\frac{1}{2}\,\frac{-4A\ln(t)+5A\ln(t)^2-2B+5B\ln(t)+5C}{t^{5/2}}=\frac{\ln(t)^2}{t^{5/2}} \quad \textbf{(29)}$$

```
> solve({-5/2*A=1, 2*A-5/2*B=0, B-5/2*C=0});
```

$$\left\{A=-\frac{2}{5},\ B=-\frac{8}{25},\ C=-\frac{16}{125}\right\} \quad \textbf{(30)}$$

Thus $\int (\ln(t))^2 t^{-\frac{7}{2}}\, dt = -\frac{2}{5}(\ln(t))^2\, t^{-\frac{5}{2}} - \frac{8}{25}(\ln(t))\, t^{-\frac{5}{2}} - \frac{16}{125} t^{-\frac{5}{2}} + C$

```
> int((ln(t))^2*t^(-7/2),t);
```

$$-\frac{2}{5}\,\frac{\ln(t)^2}{t^{5/2}}-\frac{8}{25}\,\frac{\ln(t)}{t^{5/2}}-\frac{16}{125\,t^{5/2}} \quad \textbf{(31)}$$

Side question: How would you go about solving this problem in Calculus II?

Simply use the u-substitution $u = \ln(t)$ (so $e^u = t$ and $e^u du = dt$) and so

$\int (\ln(t))^2 t^{-\frac{7}{2}} \, dt = \int u^2 \left(e^u\right)^{-\frac{7}{2}} e^u \, du = \int u^2 e^{-\frac{5}{2}u} \, du$ then do integration by parts twice and sub-back in t's.